
ASPs - Cookies, sessions, and database Access


Intensive ASP

· Creating pages on the server

· The problem of stateless dialogues

· Cookies in ASPs

· The ASP session object 

· Database access

· A larger example (in the tutorial exercises)

ASP example 3 - Create Page 
This example creates an HTML page on the server for the user.  It demonstrates use of the Request and Response objects and the File Access component.

Architecture

ASP example 3 - Create Page (continued)

ASP example 3 - Create Page (continued)
Code for CreatePage.htm

<html><head><title>CreatePage.htm</title>

<style type="text/css">

 body {

  background-color: #99cc99;

  color: #006600;

  font-family: Verdana, Arial, Helvetica, sans-serif

 }

</style>

</head>

<body>

<p>Create your own page on SouperServe</p>

<form method="post" action="CreatePage.asp">

<table width="75%" border="0">

<tr>

<td width="36%">Name:</td>

<td width="64%"><input type="text" name="name"> </td>

</tr><tr>

<td width="36%">Page colour scheme:</td>

<td width="64%">

<select name="scheme">

<option value="shock" selected>Shocking</option>

<option value="tasteful">Tasteful</option>

<option value="serious">Serious</option>

</select> 

</td></tr>

<tr>

<td width="36%">Email address:</td>

<td width="64%">

<input type="text" name="email" size="20" maxlength="20"> </td>

</tr>

ASP example 3 - Create Page (continued)
Code for CreatePage.htm (continued)

<tr><td width="36%">Favourite soup:</td>

<td width="64%">

<input type="text" name="soup" size="20" maxlength="20"> </td></tr>

<tr>

<td width="36%">Brief message:</td>

<td width="64%">

<textarea name="message" rows="2" cols="30"></textarea> </td></tr>

</table>

<p>

<input type="submit" name="Submit" value="Create Page"></p>

</form>

</body>

</html>
ASP example 3 - Create Page (continued)
Code for CreatePage.asp 

<% @LANGUAGE=VBScript %>

<% Option Explicit %>

<% ' CreatePage.asp %>

<% 

Function getVirtualDir ()

  Dim posLastSlash, virtualDir

  posLastSlash = InStrRev(request("script_name"), "/")

  virtualDir = left(request("script_name"), posLastSlash)

  getVirtualDir = virtualDir

End Function
   If isEmpty(request("name")) Then ' can't process 

                                                            ' so redirect back

       response.redirect("CreatePage.htm")

   End if

   Dim fso, foldobj, pagesDirectory, filePath, fileName

   Dim c, created, fileobj

   Set fso = Server.CreateObject( "Scripting.FileSystemObject" )

   pagesDirectory = Server.MapPath("SoupPages\")

   Set foldobj = fso.GetFolder(pagesDirectory)

   If foldobj.Size > 50000 Then ' don't let the disk fill

       Response.write("<H2>sorry can't create page - server 

full</H2>")

       Response.end

   End if
ASP example 3 - Create Page (continued)
Code for CreatePage.asp (continued)
   c = 0 ' count to ensure file names are unique

   Do 

      fileName = Request("name") & c & ".htm"  

      filePath = pagesDirectory & "\" & fileName

      If fso.FileExists(filePath) <> True Then

          fso.CreateTextFile( filePath )

          created = True

      Else

          c = c + 1       

      End If

   Loop Until created = True
' file has now been created - so open and write to it

' open the text file

   Set fileobj = fso.OpenTextFile( filePath, 2, True )

   fileobj.WriteLine("<html><head><title>")

   fileobj.WriteLine(Request("name") & "'s SouperServe page")

   fileobj.WriteLine("</title></head>")

   Dim bgcolour, textcolour, linkcolour 

   Select Case Request("scheme")

      Case "shock"

           bgcolour = "orange"

           textcolour = "lime"

           linkcolour = "blue"

      Case "tasteful"

           bgcolour = "navy"

           textcolour = "white"

           linkcolour = "red"

ASP example 3 - Create Page (continued)
Code for CreatePage.asp (continued)
      Case "serious"

           bgcolour = "gray"

           textcolour = "white"

           linkcolour = "blue"

   End Select

   fileobj.WriteLine("<body bgcolor=" & bgcolour & _

                          " text=" & textColour & _

                          " line=" & linkColour & ">")

   fileobj.WriteLine("<h2>" & Request("Name") & "'s 

SouperServe page</h2>")

   fileobj.WriteLine("<p>My favourite soup is " & 

Request("Soup") & "</p>")

   fileobj.WriteLine("<p>" & Request("Message") & "</p>")

   fileobj.WriteLine("<p>If you'd like to talk soupy")

   fileobj.WriteLine("<a href=mailto:" & Request("Email") & 

">email me</a>")

   fileobj.WriteLine("</body></html>")

   fileobj.Close

   Dim pageURL

   pageURL = "http://" & 

 


Request.ServerVariables("SERVER_NAME") & _

              

getVirtualDir() & "SoupPages/" & fileName 

%>
ASP example 3 - Create Page (continued)
Code for CreatePage.asp (continued)
<html><head><title>Create a SoupServe page</title>

<style type="text/css">

 body {

  background-color: #99cc99;

  color: #006600;

  font-family: Verdana, Arial, Helvetica, sans-serif

 }

</style>

</head>

<body>

<p><b>SouperServe page successfully 

  created </b>

<p>Your page has been created 

  - its URL is: <a href="<%=pageURL%>"> <%=pageURL%></a> 
</body>

</html>

ASP example 3 - Create Page (continued)
The basic logic of CreatePage.asp is:

· Check that request appears to have come from CreatePage.htm - if not redirect the user to it.

· Check that the SoupPages subdirectory where the HTML file is to be put doesn't take up too much space. This is to prevent repeated requests from filling up the disk.  If the limit is exceeded then output an error message and don't process the script any further.

· Make up a name for the HTML file.  The file name will be <username>n.htm where <username> is the name entered on the form and n is a number to make the file name unique.

· Write output to the new HTML file using the data entered by the user into the form.

· Send a page of output to the browser containing a link to the new HTML file.

ASP example 2 - Create Page (continued)

Things to note about CreatePage.asp

· If isEmpty(request("name")) Then 

response.redirect("CreatePage.htm")

End if

isEmpty is a VBScript function that checks whether a variable has been initialised or not.

It's not checking to see if the user entered a value into the name field but whether the request actually comes from CreatePage.htm.  How could it not?

· Set fso = Server.CreateObject( "Scripting.FileSystemObject" )

Create a FileSystemObject - the top level object that enables access to the server's file system.  
· pagesDirectory = Server.MapPath("SoupPages\")

SoupPages is a subdirectory of the directory where the ASP file resides on the server in which all the HTML pages created for users are stored.  Server.MapPath returns the absolute path to that directory (e.g. "C:\inetpub\users\SoupPages") which is needed by the code which follows.

NOTE that the SoupPages subdirectory must already exist or the application will fail.

ASP example 3 - Create Page (continued)

Things to note about CreatePage.asp (continued)
· Set foldobj = fso.GetFolder(pagesDirectory)

    If foldobj.Size > 50000 Then ' don't let the disk fill

         Response.write("<H2>sorry can't create page - server 

full</H2>")

         Response.end

    End if

Create a folder object (another object of the File Access component) and use its Size property to check that not too many HTML files have been created.  If the limit is exceeded then use Response.write to output a brief error message.

Response.end - sends the contents of the output buffer and terminates processing of the script. 

· c = 0 ' count to ensure file names are unique

Do 

      fileName = Request("name") & c & ".htm"  

      filePath = pagesDirectory & "\" & fileName

      If fso.FileExists(filePath) <> True Then

          fso.CreateTextFile( filePath )

          created = True

      Else

          c = c + 1       

      End If

Loop Until created = True
Create a file name by concatenating a number onto the name the user entered into the form, check to see if the file already exists and if it does increment the number and repeat until a unique file name is generated.

ASP example 3 - Create Page (continued)

Things to note about CreatePage.asp (continued)
· Set fileobj = fso.OpenTextFile( filePath, 2, True )
Open the file ready to write to it.  The 2 means open the file for write access (1 means read and 8 means append).  The true means create the file if it doesn't already exist.

· fileobj.WriteLine("<html><head><title>")

 fileobj.WriteLine(Request("name") & "'s SouperServe page")

etc

fileobj.WriteLine("</body></html>")

fileobj.Close

Generate the HTML code and store it in the new file.

ASP example 3 - Create Page (continued)

Things to note about CreatePage.asp (continued)

· pageURL = "http://" & 

Request.ServerVariables("SERVER_NAME") & _

              getVirtualDir() & "SoupPages/" & fileName
Make up a string containing the URL of the newly created HTML page (e.g. http://www.thing.com/asps/SoupPages/brian2.htm). 

The code assembles URL from 5 parts

· "http://"
· Request.ServerVariables("SERVER_NAME") - the server host name e.g. www.thing.com
· getVirtualDir() - the directory the ASP code is in (getVirtualDir() is defined earlier in the code) e.g. /asps/
· "SoupPages/" - the subdirectory in which the user's pages live

· fileName - the name of the newly create file e.g. brian2.htm
and finally….

· Your page has been created 

  - its URL is: <A HREF="<%=pageURL%>"> <%=pageURL%></A>
Embeds a link to the new page in the response generated for the browser e.g. 

<A HREF=

"http://www.thing.com/asps/SoupPages/brian2.htm"> http://www.thing.com/asps/SoupPages/brian2.htm</A>
 Holding a dialogue on the web is a problem

A set of interactions between a user and an application (not necessarily on the web) can be considered as a session e.g. 

During the session the software (e.g. on the ATM machine in this case) keeps track of the current state of the dialogue  (e.g. it remembers which customer it is dealing with).  

The software does this by maintaining state information (e.g. customer id etc).

The problem on the web is that HTTP is a stateless protocol - between each request from the client, the server forgets who the client is.

Trying to hold a dialogue with a server-side application is like trying to have a conversation with someone with extreme short-term memory problems ("Who are you again dear?")

Solutions to the statelessness problem

Basically there are two approaches

· Send all state information backwards and forwards each time

· Store the state information on the server and send an identifier to it with each request 

Send all state information each time

Like a memory game

· I'm going on holiday and I'm taking a camera

· I'm going on holiday and I'm taking a camera and a snorkel

· I'm going on holiday and I'm taking a camera, a snorkel and my pet octopus

· etc etc 

e.g.






Solutions to the statelessness problem (continued)

Techniques for sending all information each time:

· Cookies

· Advantages - well supported, can persist across visits to the site

· Disadvantages - some users can't/won't accept cookies

· Hidden fields in forms

<INPUT TYPE=HIDDEN  
· Advantages - easy, user need not be aware of what’s happening

· Disadvantages - only works across a series of form submits

· URL rewriting (aka URL munging)

The server-side application appends the information to the URLs of links that it generates e.g. 

<A HREF="checkout.cgi?contents=camera,60,snorkel,30">

Proceed to checkout</A>

· Advantages - user need not be aware of what’s happening

· Disadvantages - tricky to code, only lasts while user stays on site

In all of these (especially the last two) there is potential for the user to interfere with the data (e.g. code their own form containing different hidden data) unless it is encrypted.

Solutions to the statelessness problem (continued)
Store the state information on the server and send an identifier to it with each request 

· At the start of the dialogue the server application allocates a unique id and sends it to the client

· Server application stores state information (e.g. current contents of the user's shopping trolley) at the server end identified by the client id.

· Every time the client makes a request during the session it sends its id so that the server can retrieve the appropriate state information.







Solutions to the statelessness problem (continued)
Advantages of the id plus state maintained at the server end approach include:

· the state data can't be tampered with by the user

· the size of messages to and from the server can be reduced in size

· it's flexible - e.g. the data could be stored over several visits to the site

Disadvantages include:

· resources are tied up at the server end 

· how long does the state information get stored - web site visitors can leave without saying goodbye

· still requires that the id is sent with each request (usually achieved using cookies)

Saving state in ASPs

You can code ASPs to use any of the solutions to the statelessness problem described above.  The two methods we'll cover are:

· use of cookies

· use of the session object

Cookies in ASPs

Set via the Cookies collection of the Response object

[A collection is like an array but can conveniently be accessed via a key name as well as an index position.]

e.g.
Response.Cookies("favouriteColour") = "purple"
sets a cookies called "favouriteColour" to the value "purple"

Values for the other standard cookie attributes (domain, expires, path and secure) can also be set e.g.

Response.Cookies("favouriteColour").Domain = "gre.ac.uk"
Response.Cookies("favouriteColour").Expires = 

#22/01/2002#
ASPs also provide a useful facility to deal with "cookie dictionaries" whereby a single cookie can store a set of named values.

Cookies in ASPs (continued)

Cookie example




Cookie example (continued)

This mini-application consists of two ASPs which are related to each other as shown below.








Even with such a trivial example the connections and routes through the application can start to become confusing.  It's helpful to draw some to sort of diagram to illustrate the overall design of the application.

Cookie example (continued)

Cookieseg1.asp


<% @LANGUAGE=VBScript %>

<% Option Explicit %>

<% 

   If not isEmpty(Request("thename")) Then

        Response.Cookies("cthename") = Request("thename")

   End if %>
<html><head><title>ASP cookie example</title></head>

<body>

<h3>Example of using cookies - page 1</h3>



<% If isEmpty(Request("thename")) Then %>

<form method="post" action="Cookieseg1.asp">

<h3>Please enter your name: </h3>

<p><input type="text" name="thename" size="30"></p>
<p><input type="submit" name="Submit" value="Send name"></p>

</form>


<% Else %>

<h3>Welcome <%=Request("thename") %></h3>

<form method="post" action="Cookieseg2.asp">

<p><input type="submit" name="Submit" value=

"Next page"></p>

</form>

<% End if %>

</body></html>

Cookieseg1.asp (continued)

This example demonstrate a common but rather confusing style of ASP coding.

· One document may display two different HTML pages 

· If this is the first time into the page and the form has not yet been submitted:



If the form has been submitted




Cookieseg1.asp (continued)

A cookie must be set before any output is sent to the browser.  This is because command to the browser to set a cookie is part of the HTTP header.

<% @LANGUAGE=VBScript %>

<% Option Explicit %>

<% If not isEmpty(Request("thename")) Then

          Response.Cookies("cthename") = Request("thename")

     End if %>

Cookie example (continued)
Cookieseg2.asp 

<% @LANGUAGE=VBScript %>

<% Option Explicit %>


<%   If Len(Request.Cookies("cthename")) = 0 Then

           Response.redirect("Cookieseg1.asp")

        Else %>


<html><head><title>ASP cookies example - page 2</title>

</head><body>
<h3>Example  of using cookies - page 2</h3>

<p>

<%=Request.Cookies("cthename")%>'s page 

<% End if %>

</body></html>

The ASP session object
Cookies are available to all types of server-side program (e.g. CGI scripts).  

The session object is special to ASPs.  It's Microsoft's implementation of the second method of dealing with the statelessness problem i.e. "Store the state information on the server and send an identifier to it with each request"
You could take a similar approach using CGI scripts it's just that Microsoft have already done much of the hard programming by creating the convenient session object for you to use.

What is an ASP session?

· It starts when a user accesses an ASP page within a particular application (i.e. a directory on the server)

· It ends when:

· an ASP pages explicitly ends the session (Session.Abandon)

· or

· the session times out because the user hasn't interacted with the application for a certain period of time (by default 20 minutes)

The ASP session object (continued)
It's very easy to store and retrieve items from the session object e.g.

Session("sname") = "Fred"

Sets a session variable called sname to the value "Fred" 

<p><%=Session("sname")%>'s page
Outputs the value of a session variable called sname

Each client currently interacting with an ASP application will have their own session object.









The ASP session object (continued)
A session object example

The cookie example converted to use the session object



The ASP session object (continued)

Session1.asp
<% @LANGUAGE=VBScript %>

<% Option Explicit %>

<html><head><title>ASP session example</title></head>

<body><h3>Example of using the session object - page 1</h3>



<% If isEmpty(Request("name")) Then %>

<form method="post" action="Sessioneg1.asp">

<h3>Please enter your name: </h3>

<p><input type="text" name="name" size="30"></p>

<p><input type="submit" name="Submit" value=

"Send name"></p>

</form>


<% Else %>

<h3>Welcome <%=Request("name") %></h3>

<form method="post" action="Sessioneg2.asp">

<p>

<input type="submit" name="Submit" value="Next page"></p>

</form>



<%    

       Session("sname") = Request("name")

 End if %>
</body>

</html>

Sessioneg1.asp (continued)


Sessioneg2.asp

<% @LANGUAGE=VBScript %>

<% Option Explicit %>

<% If isEmpty(Session("sname")) Then

        Response.redirect("Sessioneg1.asp")

   Else 

%>

<html><head><title>ASP session example - page 2</title>

</head><body>

<h3>Example   of using the session object - page 2</h3>

<%=Session("sname")%>'s page 

<% End if %>
</body></html>
The ASP session object (continued)
The horrible truth about the session object revealed!

· It uses cookies.

· It can slow down applications

The session object uses cookies






· Most of the session data associated with a session is stored at the server end.  

· How does the server know which set of session data applies to a particular client session?

· It allocates a SessionID to the user at the beginning of the session.  This is sent to the client as a cookie.  

· The cookie is then sent to the server each time the client makes a request.

· So the session object won't work if the client doesn't accept cookies

The ASP session object (continued)

It can slow down applications 

Extract from advice about sessions from Microsoft (ASP Best Practices by George V Reilly)

· Seductively convenient but problematic

· Useful for shopping baskets

· Hampers scalability

· Wastes memory

· Session state doesn’t persist to disk 

· Sessions time out

· Requires cookies to be enabled on user’s browser

· Alternatives

· Cookies

· Encode state directly => easy, small, insecure

· ID for back-end database 

· URL rewriting

· Hidden FORM variables

So should you use the session object?

The ASP session object (continued)

TrolleySess.asp - a primitive shopping trolley example




The ASP session object (continued)

TrolleySess.asp - a primitive shopping trolley example

<% @LANGUAGE=VBScript %>

<% Option Explicit %>

<html><head><title>ASP session example - Trolley</title></head>

<body><h3>Session object - shopping trolley</h3>

<form method="get" action="TrolleySess.asp">

<h3>Choose an item:</h3>

<p><input type="radio" name="choice" value="camera"> 

camera</p>

<p><input type="radio" name="choice" value="snorkel"> 

snorkel</p>

<p><input type="radio" name="choice" value="octopus"> 
octopus</p>

<p><input type="radio" name="choice" value="cuddly toy"> 
cuddly toy</p>

<p><input type="submit" name="Submit" value="Submit"></p>

</form>


<% 

If not isEmpty(Request("choice")) Then

    Session("trolleyContents") = Session("trolleyContents") & _

                               "<li>" & Request("choice") & "</li>" 

End if

If isEmpty(Session("trolleyContents")) Then

   Response.write("<p>Trolley currently empty</p>")

Else

   Response.write("<p>Trolley contains:</p>")

   Response.write("<ul>" & Session("trolleyContents") & "</ul>")

End if
%>

</body></html>

Database access

Many server-side application require access to a database e.g. a product catalogue, customer details, accounts details

Some technologies involved

ODBC (Open DataBase Connectivity)

· Allows access to a range of types of databases without having to worry about the exact type (e.g. Access, Oracle, SQL Server etc).

OLE-DB

· The latest data access technology from Microsoft.

· Faster than ODBC

· Also allows access to non-database type data (e.g. excel spreadsheets)

· Can work with ODBC database drivers and the newer OLE-DB data providers

ADO (ActiveX Data Objects)

· The objects that you as the programmer use to interact with ODBC/OLE DB databases.  

· Not just for use with ASPs but also VB and other Microsoft languages.

Database access (continued)

Relationship between the technologies when using OLE-DB









· To a large extent the programmer isn't aware of the middle level - you feel that that your ASP code is using the ADO objects to interact directly with the actual database.

· The main place where you need to worry about what type of database you're dealing with is in establishing a connection to it.

Database access (continued)
dbExample.asp - recommended books database example


The Access database (only the books table is used in this example)


Database access (continued)
The book table.
The fields are:

· bookID

· title

· pubdate

· edition

· coverart

· authors

· publisher

dbExample.asp

<% @LANGUAGE=VBScript %>

<% Option Explicit %>

<% ' dbExample.asp %>

<html><head><title>Simple ASP example of DB access</title>

</head>

<body><h2>Recommended Books</h2>

<table border="1">

<%

   Dim dbConn, dbQuery, booksRS


   Set dbConn = Server.CreateObject( "ADODB.Connection" )

   dbConn.Open "Provider=Microsoft.Jet.OLEDB.4.0;" &_

    "Data Source=C:\dbs\bookRecommendations.mdb;"

  


   dbQuery = "SELECT * FROM books ORDER BY title;"


   Set booksRS = Server.CreateObject( "ADODB.Recordset" )

   booksRS.Open dbQuery, dbConn
dbExample.asp (continued)

   booksRS.MoveFirst()


   ' Begin a Do/While loop that will continue to

   ' iterate as long as there are records

   Do While Not booksRS.EOF

      Response.write("<tr>")

      Response.write("<td>" & booksRS("title") & "</td>")

      Response.write("<td>" & booksRS("isbn") & "</td>")

      Response.write("</tr>")

      booksRS.MoveNext()

   Loop
   dbConn.Close()

%></table>

</body>

</html>
The use of ADO objects in dbExample.asp 

Two ADO objects are used in this example

· Connection  (id = ADODB.Connection)

Used to establish a connection to the database.

· Recordset (id = ADODB.Recordset)

Used to run an SQL query on the database and manipulate the results.


dbExample.asp (continued)
Connecting to the database

The only place that the code needs to take account of the type and location of database being accessed is when using the Open method of the Connection object

dbConn.Open "Provider=Microsoft.Jet.OLEDB.4.0;" &_

    "Data Source=C:\dbs\bookRecommendations.mdb;"
(Note - one parameter is being passed in the above example - it's just split over two lines)

Open has up to three parameters:

· Connection string - used above

· User id

· Password

· The connection string gives information about how to access the database and where it is located.

· In the above example

Provider=Microsoft.Jet.OLEDB.4.0; - means that the OLE-DB driver for Access will be used to access the database

Data Source=C:\dbs\bookRecommendations.mdb; - the physical location of the database on the server

dbExample.asp (continued)
Connecting to the database (continued)
The form of the connection string depends on the type of database and how you are to access it.  For example to connect to an Access database in your public_html directory in your J: drive you're recommended to use:

dbConn.Open "Driver={Microsoft Access Driver (*.mdb)};" &_

"DBQ=\\cms-stu-storage\user-area" &_

"\X\username\public_html\database.mdb"

where

X is the first number of your username (e.g. 5 for sj593) 
username is your username e.g. sj593

database.mdb is the name of your database

Driver={Microsoft Access Driver (*.mdb)};  means that the database will be accessed using the ODBC drive for Access

For more details including  how to connect to an Oracle database see http://cms1.gre.ac.uk/laboratories/webservers.html

dbExample.asp (continued)
Database access (continued)
Having connected to the database using the connection object you can then use a Recordset object to run a SQL query against it and process the results

   dbQuery = "SELECT * FROM books ORDER BY title;"

   Set booksRS = Server.CreateObject( "ADODB.Recordset" )

   booksRS.Open dbQuery, dbConn

Then the various methods of the Recordset object can be used to retrieve the data e.g. 

booksRS.MoveFirst()

Do While Not booksRS.EOF

      Response.write("<tr>")

      Response.write("<td>" & booksRS("title") & "</td>")
      Response.write("<td>" & booksRS("isbn") & "</td>")
      Response.write("</tr>")

      booksRS.MoveNext()

Loop

dbExample.asp (continued)
Database access (continued)

Summary of the ADO Recordset object methods and properties used in the example

	Method/Property
	Description

	Open() method
	Creates a Recordset by running an SQL query against the database

	MoveFirst() method
	Moves to the first record returned by the query

	EOF property
	This is true when all records in the Recordset have been processed and false if there are still more records to process

	MoveNext() method
	Moves to the next record returned by the query



A tiered architecture for web applications

As you can see from some of the examples we've looked at ASP code can become a messy clutter of HTML, generated HTML, application/business logic and database queries.

For this reason (and some others including performance and scalability) a tiered architecture is often recommended for web applications.  An example of such an architecture from ASP Best Practices by George V Reilly is shown below.







Separation of application logic and user interface/presentation code is encouraged/enforced in ASP.NET

A tiered architecture for web applications (continued)
Client Tier

· Code that is executed at the client end e.g. HTML and JavaScript.

· For performance reasons it's a good idea to offload as much work to this tier as possible (e.g. validation, formatting, scrolling through records).

· For browser compatibility reasons it's a good idea to keep this tier as simple as possible.

Middle Tier - Presentation Layer

· Code that is executed at the server end and is concerned with formatting and presenting the data (e.g. generating the code for an HTML table to layout records returned from a search).

· Could be ASP or use some other server side technology.

Middle Tier - Application Logic

· Code that's executed at the server end and contains the main application/business logic (e.g. credit checking, transferring money from one account to another, checking stock levels).

· This code should not need changing just because the presentation requirements change.

· Normally created as components using some component/object standard e.g. COM, Enterprise Java Beans

A tiered architecture for web applications (continued)
Data Tier

· The actual database (e.g. Oracle or SQL server).

· For performance reasons it's good to use features such as stored procedures that the DBMS can optimise.

Tutorial exercises

1. Create and run a "hello world" asp that just outputs a welcome message from the server when run e.g.


REMEMBER that the asp has to run through a web server - you can't just save the file an double click it to run.

2. Create a form which allows the user to input two numbers.  When the submit button is clicked the numbers should be sent to an ASP program (you'll need to create this too) which adds the two numbers together and displays the result e.g.



Obviously this is a very trivial application but in principle what do you think are the pros and cons of using ASPs to carry out processing such as this as opposed to doing it at the client-side?

3. A portal website decides to display a page of hot links which may be updated several times week.  Rather than edit the HTML code each time the list changes they decide to store the current set of addresses to be linked to in a plain text file called links.dat e.g.


Your job is to write the ASP page which reads this file and generates a web page containing links to the current list set of hot sites e.g.


In order to do this you will need to open the links.dat file and read it a line at a time.  The skeleton of the code to loop through the file is given below:

Do while not fileobj.AtEndOfStream 

         linkAddress = fileobj.ReadLine()

' process the link just read

Loop
4.
Make sure that you understand CreatePage.asp. Copy it to your area on cms-stu-iis (i.e. somewhere under public_html on your J: drive) and make sure that you can run it from there.  Remember that you need to create your own SoupPages folder.

Amend it so that as well as displaying a link to the new page it also displays a list of links to all the pages that exist in the SoupPages folder e.g. 


To do this you will need to loop through all the objects in the folder object's file collection.  Skeleton code is given below:

for each file in foldobj.Files

   '  construct a link using the file.name property

next
4.
Complete the following table which compares the use of cookies and the use of the session object for maintaining state.

	
	Cookies
	Session object

	How set
	using Request.Cookies e.g. Request.Cookies("id") = "AY087"


	

	How retrieved
	
	

	Persistence (i.e. how long they last)
	
	Only during the session i.e. until the session is explicitly ended or it times out.

	Ease of programming
	
	

	Possible performance implications for the server
	
	

	Security implications
	
	


5.
Rewrite TrolleySess.asp to use cookies rather than the session object.

6.
Detecting "refresh" (rather tricky don't get bogged down with this one)

If you experiment with running TrolleySess.asp you may notice that if the user clicks the browser refresh button the last item they chose will be added to the trolley again.  For example if you add an octopus to your trolley and then click refresh three times you'll end up with four octopuses.  

Try to devise some code that will prevent this from happening i.e. will not add the item again when the user clicks refresh.

7.
Amend dbexample.asp so that it includes the book authors and also displays column headings e.g.


8.
Amend dbexample.asp again so that the description of the course for which each book is recommended is displayed together with the book e.g.


You'll notice that some books appear twice in the list because they are recommended for more than one course.

The slightly tricky thing about this exercise is that book details and the course names are held in different tables in the database.





This means that the tables need to be joined in order to get at the required data.  There are two approaches to this:

a) 
In Access create a query that retrieves the required data e.g.


and then use this in the SQL statement in your ASP e.g.

dbQuery = "SELECT title, isbn, authors, description " _

             & "FROM booksandcourses"
b)
Write a more complex SQL query in your ASP code that does the join e.g. something (but not exactly) like:

dbQuery = "SELECT title, description " _

& "FROM books, recommendation, courses " _

& "WHERE (books.bookID = recommendation.bookID)" _

& " and (courses.courseID = recommendation.courseID);"
9.
Amend your answer to exercise 5 to only retrieve books for one particular course e.g.



For the moment you can do this by "hard-coding" the name of the required course in the SQL query in your ASP code e.g.

dbQuery = "SELECT title, isbn, authors, description " _

             & "FROM booksandcourses " _

             & "WHERE description = 'DIPA'"

Notice that SQL requires that literal strings (e.g. 'DIPA') are contained within quotes.

10.
Amend your answer to exercise 6 so that instead of a hard-coded course name you use a course name entered by the user into a text field on a form e.g.



The only tricky bit about this is how you get the variable data (i.e. the course name) into the SQL query.  You can pick it up easily from Request("nameoffield") but you must make sure that when it is processed as part of the query it is surrounded by quotes e.g.

"WHERE description = '" & Request("nameoffield") & "'"


11.
One more database example - this time to insert data

Amend your answer to exercise 4 to insert a record into the books table.


One way you can do this is to use the SQL INSERT statement to enter records into the database.  The syntax for this SQL statement is:

INSERT INTO nameOfTable (fieldname1, fieldname2, etc)

VALUES (value1, value2, etc)
e.g.
INSERT INTO books (Title, pubdate, edition, isbn, authors, publisher) 

VALUES ('Fun with SQL', '1935', '2nd', '1 12354354 1',

'D. Ata and B. Ase', 'Books for the Future')
The trick is to use the values entered into the form rather than hard-coded data as shown in the example above.

enter PIN number





display menu





request balance





request money





display balance





dispense money





quit





shopping trolley empty add a camera





shopping trolley contains camera, cost £60





shopping trolley contains camera, cost £60 - add a snorkel





shopping trolley contains camera, cost £60 and snorkel, cost £30





Client





Server





Client 1





Client 2





Server application





state for client 1





state for client 2





Quick Quiz Question 1





Image a web site where users can play a game of chess against a server side application (an ASP???).  What sort of state information would need to be maintained?





cookieseg1.asp





surfed or linked to directly





cookieseg2.asp





form submit


"next page"





form submit


"send name"





redirect - if cookie is empty





If name has been entered then store it as a cookie





If name has NOT been entered then prompt for it





If name has been entered then welcome the user





back to Cookieseg1.asp





to Cookieseg2.asp





If name has been stored as a cookie then use it to display a message to the user





Quick Quiz Question 2





How long do you think the cookie set in the above example will persist for?





If name has not been stored as a cookie then redirect to the page that prompts for name to be input





Quick Quiz Question 3





How is it possible that this page could be requested without the name cookie having been set?





� EMBED Word.Picture.8  ���





ASP application





session object for client 1





Client 1





session object for client 2





Client 2





If name has NOT been entered then prompt for it





If name has been entered then welcome the user





and store the name in the session object





Quick Quiz Question 4





In the cookies example the first thing the ASP script did was to store the name in the cookie.  In the session example the last thing that happens is that the script stores the name in the session object.  Can you account for this difference?








If name has not been stored in session object then redirect to the page that prompts for name to be input





If name has been stored then use it to display a message to the user





ASP application





session object for client 1





Client 1





Client 2





session object for client 2





If something has been selected then add it to the trolley





Display the trolley contents








application





ASP code using ADO objects 





OLE-DB 











Actual database e.g. calalogue.mdb created in Access





creates HTML page for the user containing information from the form





CreatePage.asp











form POST





CreatePage.htm





�





Recordset object





ID�
title�
pubdate�
edition�
isbn�
coverart�
authors�
publisher�
�
9�
Engineering Distributed Objects�
2000�
1�
0-471-98657-7�
�
Emmerich�
Wiley�
�
8�
Internet and the World Wide Web�
2000�
1�
0-13-016143-8�
�
Deitel, Deitel and Nieto�
Prentice Hall�
�
7�
Java : How to Program�
1999�
3�
0-13-012507-5�
jhtp3.gif�
Deitel and Deitel�
Prentice Hall�
�
10�
Java in a Nutshell�
1999�
3�
1-56592-487-8�
�
Flannagan�
O'Reilly�
�





















ASP code





Database





Connection





Recordset - query and results





Quick Quiz Question 6





How would the code of dbexample.asp need to be changed to make it display the book publisher as well as the title and isbn?








books





bookID


title


isbn


authors


etc





Data


Tier





DBMS (Access, SQL Server, Oracle etc)MS





Application


Logic





Components





(VB, C#, Window Script Tool, EJBs, etc)





Presentation


Layer





ASP





(or PHP, Perl, JSP etc)





Middle Tier -- ASP











Client


Tier





Browsers


(HTML, JavaScript etc)





recommendation





courseID


bookID


etc





courses





courseID


description





etc





single quotes surrounded the data held in Request("nameoffield")





ODBC


driver  











OLE-DB


data provider for Access








OLE-DB


data provider for Oracle








OLE-DB


data provider for SQL Server








create an ADO Connection object and use its Open method to open the database for use.  





Create the SQL query that will be run against the database 





Create an ADO Recordset object and use it to run the query against the database - the results of the query are stored in the Recordset





Use the MoveFirst() method of the Recordset object to position on the first row/record of data returned by the query





Loop through the results of the query extracting the title and isbn fields and displaying them





SoupPages


gill1.htm


fred0.htm


fred1.htm








Get data entered by the user into the form





Create a page for the user in the SoupPages subdirectory 





Request





CreatePage.asp





Response





Return confirmation the user with a link to their page

